V olimpiada

I etapas

45. Standartiniai numeriai. Pradiniai duomenys – knygos numeris – pavaizduoti simboliniu masyvu. Pirmiausia simbolius perkoduojame skaitmenimis:

         if simb in ['0'..'9']

            then sk := ord (simb) - ord ('0');

Užrašydami simbolius skaitmenimis, juos iš karto dauginame iš reikiamo daugiklio ir sumuojame (šią sumą pavadinkime sum). Atskirai įsidėmime trūkstamo skaitmens t daugiklį d bei kontrolinį skaitmenį k. Visa tai galime užrašyti šitaip:

(sum + t ( d) mod 11 = k.

Būtų galima bandyti suprastinti reiškinį, tik kažin ar verta tai daryti. Ilgai tektų vargti taikant formules, o sutaupytume nedaug. Kadangi žinoma, jog trūksta vieno skaitmens, tai pakanka perrinkti visus variantus nuo 0 iki 9 ir rasti tinkamą. Radus tinkamą skaitmenį, reikia nepamiršti jį užrašyti atitinkančiu simboliu, kaip to prašoma sąlygoje.

Jei patikrinę visus variantus, tinkamo skaitmens nerandame, rezultatui priskiriame simbolį N.

const N = 10;    { ISBN numerio skaitmenų skaičius }

type numeris = array [1..N] of char;



function trūkstamas (s: numeris): char;

  var d,      { daugiklis prie nežinomo skaitmens }

      k,      { kontrolinis skaitmuo }

      i, sum: integer;

      rasta: boolean; { ar rastas tinkamas skaitmuo }

begin

  sum := 0;

  for i := 1 to N-1 do

    if s[i] in ['0'..'9']

       then sum := sum + i * (ord(s[i]) - ord('0'))

       else d := i;

  if s[N] = 'X'

     then k := 10

     else k := ord(s[N]) - ord('0');

  rasta := false;

  i := 0;

  { ieškoma trūkstamo skaitmens }

  while not rasta and (i <= 9) do

    if (sum + i * d) mod 11 = k

       then rasta := true

       else i := i + 1;

  if rasta

     then trūkstamas := chr (i + ord('0'))

     else trūkstamas := 'N'

end; { trūkstamas }

46. Sudėtis. Pradiniai duomenys – trys simbolių eilutės, sudarytos iš skaitmenų ir žvaigždučių (*). Kai reikės sudėti skaičius, bus nepatogu, jei tarp jų maišysis žvaigždutės, todėl skaitant pradinius duomenis geriau jas paversti skaitiniais dydžiais, pavyzdžiui, -1. 

Vienas ar abu dėmenys gali būti trumpesni nei suma. Jau skaitydami pradinius duomenis, trūkstamą pradžią papildome nuliais. Tačiau reikia nepamiršti įsiminti, kiek skaitmenų turi kiekvienas dėmuo. To nepadarius gali būti negerai, jei nežinomas kurio nors iš dėmenų pirmasis skaitmuo. Mat nė vienas iš dėmenų, taip pat ir suma, negali prasidėti nuliu. Jei neįsiminsime skaitmenų skaičiaus, radus sprendinį, kuriame dėmuo prasideda nuliu, nebus aišku, iš kur tas nulis: ar jis reiškia, kad nepavyko rasti sprendinio, ar tiktai kad dėmuo yra trumpesnis.

46.1 pav.

Bandome sudaryti algoritmą. Pirmiausia pastebime, kad patogiau įsiminti ne dėmens ilgį, o ties kuriuo skaitmeniu prasideda dėmuo.

Jei stulpelyje trūksta tik vieno skaitmens, visuomet skaičius galime atstatyti vienareikšmiškai (arba pasakyti, kad sprendinio nėra). 

46.2 pav.

Jei stulpelyje nežinomi du skaitmenys, tuomet teks atlikti perrinkimą: vietoj vieno iš dviejų nežinomų skaitmenų reikės bandyti įstatyti visus skaitmenis nuo 0 iki 9. Analogiškai elgiamasi, kai nežinomi visi trys skaitmenys: teks bandyti perrinkti du iš trijų skaitmenų. 

Bus patogiau, jei visus nežinomus skaitmenis, kuriuos reikės perrinkti, surašysime į lentelę. 

46.3 pav.

Tada teliks pritaikyti nesudėtingą rekursiją: vietoj pirmo nežinomo skaitmens bus mėginama įrašyti visus skaitmenis nuo 0 iki 9, atlikus tai, skaitmenis įrašyti bus mėginama vietoje antro nežinomo skaitmens ir t. t. Kai vietoj nežinomų įstatysime visus skaitmenis, mėginsime atstatyti likusius skaitmenis (dabar kiekviename stulpelyje trūks ne daugiau kaip vieno skaitmens). Jei atstatyti pavyks, puiku, skaičiai rasti, jei ne, turėsime grįžti žingsniu atgal ir toliau tęsti darbą.

const M = 30;       { maksimalus nežinomų skaitmenų skaičius }

      ILGIS = 10;   { maksimalus skaičiaus ilgis }

type skaičiai = array [1..3, 1..ILGIS] of -1..9; 

                         { -1 žymime nežinomą skaitmenį }

     nežinomi = array [1..M] of

                   record

                      ska, skt: integer

                      { kurio skaičiaus kuris skaitmuo nežinomas }

                   end;



{ žemiau surašyti globalieji kintamieji }

var sk: skaičiai;   { trys duotieji skaičiai }

    než: nežinomi;  { nežinomų ir iš karto neatstatomų skaitmenų sąrašas }

    kiekis,         { ir jų kiekis }

    prad2, prad1,   { ties kuriuo skaitmeniu prasideda pirmas ir antras skaičius }

    ilg: integer;   { ilgiausio skaičiaus ilgis }

    yra: boolean;   { ar rastas sprendinys }



procedure reikšmė;

{ apskaičiuoja reiškinio reikšmę, kai kiekviename stulpelyje trūksta }

{ ne daugiau kaip vieno skaitmens }

  var kopija: skaičiai;

      min,  { kiek saugoma „mintyje“ }

      i: integer;

      gerai: boolean;

begin

  { įstatysime trūkstamus skaitmenis }

  kopija := sk;

  min := 0;

  gerai := true;

  for i := ilg downto 1 do

   begin

    if sk[1, i] = -1  { pirmasis skaitmuo nežinomas }

    then 

      if prad2 > i

      then sk[1, i] := sk[3, i] + min

      else sk[1, i] := (sk[3, i] - min - sk[2, i] + 10) mod 10;

      if sk[2, i] = -1  { antrasis skaitmuo nežinomas }

      then if prad1 > i

           then sk[2, i] := sk[3, i] + min

           else sk[2, i] := 

                  (sk[3, i] - min - sk[1, i] + 10) mod 10;

      if sk[3, i] = -1  { trečiasis skaitmuo nežinomas }

      then sk[3, i] := (sk[1, i] + min + sk[2, i]) mod 10;

      { ar pavyko sėkmingai atstatyti }

      if  (sk[1, i] < 0) or (sk[2, i] < 0) or (sk[3, i] < 0) or

         (((sk[1, i] + sk[2, i] + min) mod 10) <> sk[3, i])

      then gerai := false;

      min := (sk[1, i] + min + sk[2, i]) div 10

   end;

  { patikrinsime, ar nė vienas skaičius neprasideda nuliais }

  yra := gerai and (sk[1, prad1] <> 0) and (

         sk[2, prad2] <> 0) and (sk[3, 1] <> 0);

  if not yra

     then sk := kopija

end; { reikšmė }



procedure spręsti (kuris,   { į kurį skaitmenį įstatysime  ir }

                   skaitmuo: integer  { kokį skaitmenį });

{ rekursinė procedūra, įstatinėjanti nežinomus skaitmenis }

  var s: integer;

begin

  if kuris = 0    { įstatyti visi skaitmenys }

     then reikšmė { apskaičiuojame reikšmę }

     else begin

            { įstatysime i-ąjį skaitmenį }

            sk[než[kuris].ska, než[kuris].skt] := skaitmuo;

            { įstatysime į likusius nežinomus skaitmenis }

            for s := 0 to 9 do

              if not yra

                 then spręsti (kuris-1, s)

          end

end; { spręsti }



function nežin (sk: skaičiai; i: integer): integer;

{ randa, kiek i-ame stulpelyje yra nežinomų skaitmenų }

  var kiek: integer;

begin

  kiek := 0;

  if sk[1, i] = -1

     then kiek := kiek + 1;

  if sk[2, i] = -1

     then kiek := kiek + 1;

  if sk[3, i] = -1

     then kiek := kiek + 1;

  nežin := kiek

end; { nežin }



procedure sudėti (var sk: skaičiai; { duotieji skaičiai }

                  var yra: boolean  { ar yra sprendinys });

  var kiek, { nežinomų skaitmenų skaičius stulpelyje }

      i, s, nr: integer;

begin

  kiekis := 0; { tiek yra iš pirmo karto neatstatomų skaičių }

  { rasime nežinomų ir iš karto neatstatomų skaitmenų vietas }

  for i := 1 to ilg do

    begin

      kiek := nežin(sk, i);

      if (kiek > 1)

         then begin { jei stulpelyje daugiau nei vienas nežinomas skaičius, }

                { įtraukiame pirmą nežinomą skaitmenį }

                kiekis := kiekis + 1;

                if sk[1, i] = -1

                   then nr := 1

                   else nr := 2;

                než[kiekis].ska := nr;

                než[kiekis].skt := i;

                if kiek = 3  { jei reikia, įtraukiame antrą skaitmenį }

                   then begin

                          kiekis := kiekis + 1;

                          než[kiekis].ska := nr;

                          než[kiekis].skt := i

                        end

              end

    end;

  yra := false;

  for s := 0 to 9 do

    if not yra

       then spręsti (kiekis, s);

end; { sudėti }



Testo nr.�Pradinis duomuo�Rezultatai�Komentarai��1�1487*

2*124

38**5�14871

24124

38995�Visuose stulpeliuose trūksta ne daugiau kaip vieno skaitmens��2�1*5793*

14*24*6

291**33�1457937

1452496

2910433�Sumuojant susidaro skaičius „mintyje“��3�   102*5

5*81*141

57*2038*�   10245

57810141

57820386�Pirmas dėmuo trumpesnis už likusius��4� 541

 6*4

1*9*� 541

 654

1195�Abu dėmenys trumpesni už sumą��5�12*45

*23*1

1464*�Nėra sprendinio�Sprendinys būtų tik tada, jei dėmuo galėtų prasidėti nuliu��6�...*4*8*1*

58*1*8**4*

***00*0*54�   8408010

5801682944

5810090954�Sudėtingesnis testas��7�****

****

****�1000

8999

9999�Sudėtingesnis testas��

47. Žirgo kelias. Šis uždavinys panašus į 18 uždavinį (Paprastas kelias). Sprendimų idėjos nedaug skiriasi, tačiau ją surasti sunkiau.

Pirmiausia susitarkime, kad šachmatų lenta žymima įprastomis koordinačių plokštumos koordinatėmis. Lenta didelė – taip patogiau, pačiam uždaviniui tai neturi reikšmės. 

Pastebėsime, kad abiejų žirgų koordinatės mums apskritai nerūpi – domina tik jų skirtumai. Suradę šiuos skirtumus (x ir y ašių atžvilgiu), išskiriame du atvejus:

1) kai abu skirtumai patenka į intervalą [0; 4];

2) kai skirtumai (ar bent vienas jų) didesnis už 4.

Pirmuoju atveju, išbandę ėjimus rankiniu būdu, sudarome lentelę (tariame, kad pirmasis žirgas yra langelyje (0, 0):

4�2�3�2�3�4��3�3�2�3�2�3��2�2�1�4�3�2��1�3�2�1�2�3��0�0�3�2�3�2���0�1�2�3�4��

Antruoju atveju reikia eiti tol, kol abu skirtumai taps mažesni už 4, t. y. pasieksime lentelėje išnagrinėtą situaciją.

Kaip turi eiti žirgas, kad jo kelias būtų trumpiausias?

Logiška samprotauti šitaip. Kurios koordinatės skirtumas didesnis, tą mažiname 2 (kitą mažinsime 1, nes žirgo ėjimas yra „L“ raidės pavidalo). Šitaip darydami, priartėsime prie pirmame atvejyje gautos lentelės – tuomet beliks tik pridėti iš jos likusį žirgo ėjimo skaičių. Jei vienos iš koordinačių skirtumas tapo lygus nuliui (t. y. pasiekėme reikiamą stulpelį ar eilutę), tuomet kiekvienu ėjimu žirgas šoka į gretimą eilutę (stulpelį) o tolesniu ėjimu grįžta į reikiamą eilutę (stulpelį). T. y. tos koordinatės skirtumas keičiasi tarp 0 ir 1.

Jei lenta būtų baigtinė, tuomet tektų išskirti vieną išimtį: kai abu žirgai stovi lentos kampo pakraštyje greta įstrižai. Tuomet antrasis nepasieks pirmojo per du ėjimus (negalės išeiti už krašto). Jam prireikus 4 ėjimų.

Tačiau sąlygoje nurodyta, kad lenta begalinė, tad tokios situacijos nenagrinėjame.

procedure žirgas (Ax, Ay, Bx, By: integer; { žirgų koordinatės }

                 var kiek: integer { ėjimų skaičius });

  const lent: array[0..4, 0..4] of integer = 

               { rankomis apskaičiuotoji lentelė } ((0, 3, 2, 3, 2),

                                        (3, 2, 1, 2, 3),

                                        (2, 1, 4, 3, 2),

                                        (3, 2, 3, 2, 3),

                                        (2, 3, 2, 3, 4));

  var dx, dy: integer;  { koordinačių skirtumai }

begin

  dx := abs (Ax - Bx);

  dy := abs (Ay - By);

  kiek := 0;

  while (dx > 4) or (dy > 4) do { einame mažindami dviem vienetais }

    begin                       { didesniąją koordinatę }

      if dx > dy

         then begin

                dx := dx - 2;

                dy := abs (dy - 1)

              end

         else begin

                dy := dy - 2;

                dx := abs (dx - 1)

              end;

      kiek := kiek + 1

    end;

  kiek := kiek + lent[dx, dy] { baigiame prisumuodami iš lentelės }

end; { žirgas }

48. Sklypas. Iš pirmo žvilgsnio uždavinys gali atrodyti keblokas. Juk pastatas gali būti įvairiausiai išsirangęs. Panagrinėkime sąlygoje pateiktą pavyzdį, kai n = 5. Pirmiausia imkime porą atvejų (48.2.a-b pav.):

48.2.a-b pav.

Abiem atvejais sklypo plotas bus vienodas – lygus 25. Kaip jis apskaičiuojamas? Pastato pločiu p pavadinkime stačiakampio eilučių, kuriose yra pastatą sudarantys kvadratėliai, skaičių. Pastato ilgiu i pavadinkime stačiakampio stulpelių, kuriuose yra pastatą sudarantys kvadratėliai, skaičių.

Sklypo plotas lygus:

S = (p + 1) * (i + 1).

Savaime suprantama, kad pastato plotis ir ilgis negali būti mažesni už vienetą ar didesni už n.

Jei pastato plotis lygus 1, tai jo ilgis tegali būti lygus n (48.3 pav.):

48.3 pav.

Jei pastato plotis lygus 2, tai jo ilgis negali būti mažesnis kaip n div 2 + n mod 2 langelių (48.4 pav.):

48.4 pav.

Taip pat pastatas negali būti ilgesnis kaip n - 2 + 1 langeliai. Pastaruoju atveju jis bus L raidės formos (48.5 pav.).

48.5 pav.

O jei nuo šio pastato pirmos eilės nuiminėsime po vieną kvadratėlį ir jį dėsime į antrą eilę, pamatysime, kad pastato ilgis gali būti bet koks skaičius iš intervalo �[n div 2 + n mod 2; n - 2 + 1]. 

Taigi išsėmėme visus variantus. Mat viena vertus, parodėme, kad pastato ilgis negali būti mažesnis ar didesnis už intervalo rėžius, antra vertus, tapo aišku, kad pastato ilgis gali būti bet koks skaičius iš šio intervalo. 

Išnagrinėjome atvejį, kai pastato plotis lygus 2. Sudarysime formulę, tinkančią kiekvienam atvejui. Jei pastato plotis yra p, jo ilgis visuomet bus iš intervalo

�                     1, jei n mod p <> 0 

�[n div p +                                         ,                   n - p + 1]

                     0 priešingu atveju ; 

Dabar jau nebesudėtinga parašyti algoritmą. Pradžioje pastato plotį imame lygų vienetui. Surandame tokio pločio pastato visus galimus ilgius bei sklypo plotus. Plotus rašome į masyvą. Prieš įrašant būtina patikrinti, ar šis plotas nebuvo rastas. Po to plotį didiname vienetu, iš naujo apskaičiuojame pastato ilgio intervalo rėžius ir t. t. 

Tikrinimą galime nutraukti, kai pastato plotis taps didesnis už ilgį. Mat jei skaičiuosime toliau, simetriškai gausime jau ištirtus variantus. 

const M = 1000;  { maksimalus skirtingų atsakymų skaičius }

type masyvas = array [1..M] of integer;



procedure sklyp (n: integer;

                 var plotas: masyvas; { skirtingų sklypo plotų sąrašas }

                 var kiek: integer { skirtingų atvejų skaičius });

  var p,           { pastato plotis }

      i,           { pastato ilgis }

      apat, virš,  { ilgio intervalo rėžiai }

      s,           { sklypo plotas }

      j: integer;

      rasta: boolean;

begin

  kiek := 0;

  for p := 1 to n div 2 + 1 do

    { jei p >= n div 2 + 1, pastato plotis tikrai viršys jo ilgį }

    begin

      if n mod p = 0    { randame sklypo ilgio intervalo rėžius }

         then apat := n div p

         else apat := n div p + 1;

      virš := n - p + 1;

      for i := apat to virš do

        begin

          s := (i + 2) * (p + 2); { apskaičiuojame pastato plotą }

          { tikrinsime, ar toks plotas dar nebuvo rastas }

          rasta := false;

          for j := 1 to kiek do

            if plotas[j] = s

               then rasta := true;

          { jei tokio ploto nerasta, įtraukiame jį į sąrašą }

          if not rasta

             then begin

                    kiek := kiek + 1;

                    plotas [kiek] := s

                  end

        end

    end

end; { sklyp }

II etapas

49. Lygybė. Sąlyga prašo rasti visus galimus sprendinius, taigi teks atlikti perrinkimą. Jei reikėtų gauti tik vieną sprendinį, būtų galima sugalvoti žymiai efektyvesnį algoritmą:

Tarkime, kad turime skaičių 148?2. Jį galime užrašyti 14802 + ? ( 101. Taigi kiekvienas klaustuku pažymėtas skaitmuo reiškia kurį nors dešimties laipsnį, padaugintą iš skaitmens. 

Pirmiausia įsidėmėsime (pavyzdžiui, lentelėje), kiek trūksta skaitmenų prie kiekvieno dešimties laipsnio:

Dešimties laipsnis�100�101�102�...�109�-100�-101�...�-109��Kiek trūksta skaitmenų�1�0�0�...�0�1�0�..�0��

Beje, reikia nepamiršti, kad skaitmenys gali būti nutrinti ir dešinėje lygybės pusėje. 

Po to vietoj trūkstamų skaitmenų įrašome nulius ir randame reiškinio reikšmę. Pavyzdžiui, reiškinyje

150 - 10? + ? + 3 - 47 + 4 ( 13

klaustukus pakeičiame nuliais ir randame reiškinio reikšmę:

150 - 100 +0 + 3 - 47 + 4 ( 10

Iš tikrojo rezultato atėmę suskaičiuotą reiškinio reikšmę, gausime skaičių, kurį reikia gauti iš trūkstamų skaitmenų (pateiktu atveju reikia gauti 13 - 10 ( 3). Taigi reikia gauti vienženklį teigiamą skaičių. Vieną trūkstamą skaitmenį prie 100 prilyginame 3, o visus kitus skaitmenis – nuliui. Gauname sprendinį:

150 - 100 + 3 + 3 - 47 + 4 ( 13

Pastebime, kad jei prie 100 netrūktų nė vieno skaitmens, reiškinys sprendinio neturėtų. 

Kai reikia rasti visus sprendinius, paprasčiausiai imame visus galimus nežinomų skaitmenų variantus ir bandome įstatyti juos į reiškinį. Šitaip sutvarkę visą reiškinį, toliau skaičiuojame jo reikšmę.

function teis (s: string): boolean;

{ patikrina, ar teisinga duotoji lygybė }

  var reikšmė,        { kairioji lygybės pusė }

      i, sk, kodas: integer;

      simb: string;   { skaičius }

      nul: boolean;   { nė vienas skaičius kairėje lygybės pusėje }

                      { negali būti lygus 0 }

begin

  nul := false;

  reikšmė := 0;

  simb := '';

  i := 1;

  { apskaičiuojame kairiąją lygybės pusę }

  while s[i] <> '=' do

    begin

      { atskiriame skaičių }

      while s[i] in ['0'..'9'] do

        begin

          simb := simb + s[i];

          i := i + 1

        end;

      { apskaičiuojame jo reikšmę }

      val(simb, sk, kodas);

      reikšmė := reikšmė + sk;

      nul := nul or (sk = 0);

      if s[i] <> '=' { ženklas '+' ar '-' reiškia kito skaičiaus pradžią }

         then begin

                simb := s[i];

                i := i + 1

              end

    end;

  simb := copy(s, i+1, length(s)-i+1);

  val(simb, sk, kodas);

  { sulyginame abi lygybės puses ir patikriname, ar nebuvo nulių }

  teis := (sk = reikšmė) and not nul

end; { teis }



procedure spręsti (s: string; { duotoji simbolių eilutė }

                   var yra: boolean { ar rastas nors vienas sprendinys });



  procedure skaičiuoti;

  { rekursinė procedūra, perrenkanti visus variantus }

  { globalieji kintamieji: s, yra }

    var vieta,       { simbolio „?“ vieta eilutėje }

        i: integer;

  begin

    vieta := pos ('?', s);

    if vieta > 0   { jei dar yra nežinomų skaitmenų }

    then for i := 0 to 9 do

           begin

             s[vieta] := chr(i + ord('0'));

             skaičiuoti;

             s[vieta] := '?' { atstatome buvusią reikšmę }

           end

    else if teis (s) { jei rastas sprendinys }

         then begin

                writeln (s);

                yra := true

              end

  end; { skaičiuoti }



begin

  yra := false;

  skaičiuoti

end; { spręsti }



Testo nr.�Pradiniai duomenys�Rezultatai�Komentarai��1�1000+2000-3000=?�1000+2000-3000=0�Trūksta vienintelio skaitmens dešinėje lygybės pusėje. Atstatytasis skaitmuo lygus nuliui��2�57+84=141?�LYGYBĖJE YRA KLAIDŲ�Trūksta skaitmens dešinėje lygybės pusėje��3�14?5+1000-2999=-594�1405+1000-2999=-594�Trūksta skaitmens kairėje pusėje. Atstatytasis skaitmuo lygus nuliui��4�2+3+?-2?+50=?9�2+3+1-27+50=29

2+3+2-28+50=29

...�Trūksta kelių skaitmenų, galimi 9 sprendiniai��5�200+3???-200=3222�20000+32222-20000=32222�Trūksta kelių skaitmenų, tačiau sprendinys – tik vienas��

50. Laužtė. Laužtės apribotą plotą galima nagrinėti sluoksniais (50.1 pav.):

50.1 pav.

(Einame” kiekvienu sluoksniu, kol sutinkame pirmą laužtės atkarpą tame sluoksnyje. Ja prasideda laužtės ribojama figūra. (Keliaudami” sluoksniu toliau, skaičiuojame praeitus langelius, kol sutinkame kitą laužtės atkarpą. Ties ja figūros dalis, priklausanti nagrinėjamam sluoksniui, baigėsi. Šiuos veiksmus kartojame toliau, kol nagrinėjamame sluoksnyje nebeliks laužtės atkarpų . Kaip pavyzdį imkime septintą sluoksnį (užbrūkšniuoti ploteliai yra priklauso figūrai, 50.2 pav.):

50.2 pav.

Matome, kad figūros dalies, esančios septintame sluoksnyje plotas lygus 5. Taip sluoksnis po sluoksnio ir suskaičiuojamas figūros plotas.

Galimi įvairūs būdai, kaip realizuoti šį algoritmą. Žemiau pateiktame algoritme daroma taip. Figūra į sluoksnius suskaidoma tiesėmis, lygiagrečiomis x ašiai. Todėl horizontalios laužtės atkarpos tolimesniam sprendimui nėra reikalingos. Vertikalios atkarpos, surikiuojamos koordinatės x didėjimo tvarka. Tuomet peržiūrimos kiekvienam sluoksniui priklausančios atkarpos ir skaičiuojamas figūros plotas.

const M = 100; { maksimalus laužtės taškų skaičius }

type taškas = record

                x, y: 0..maxint

              end;

     koordinatės = array [1..M] of taškas;

     atkarpos = array [1..M] of record  { atkarpos pradžia ir pabaiga }

                                  pr, pb: taškas

                                end;  { pr < pb }



procedure sudaryti_atkarpas 

               (k: koordinatės; sk: integer;

                var atk: atkarpos;  { sudarytosios atkarpos }

                var s: integer { jų skaičius });

{ sudaromos visos vertikalios atkarpos }

  var i: integer;

begin

  s := 0;

  for i := 1 to sk do

    if k[i].x = k[i+1].x     { jei atkarpa vertikali }

       then begin

              s := s + 1;

              if k[i].y < k[i+1].y  { turi galioti pr < pb }

                 then begin

                        atk[s].pr := k[i];

                        atk[s].pb := k[i+1]

                      end

                 else begin

                        atk[s].pr := k[i+1];

                        atk[s].pb := k[i]

                      end

            end

end; { sudaryti_atkarpas }



  procedure rikiuoti (var atk: atkarpos;

                      s: integer);

  { atkarpos surikiuojamos didėjimo tvarka pagal koordinatę x }

  { galima naudoti bet kurį rikiavimo algoritmą }

  { rikiuojama burbuliuko metodu, algoritmas paimtas iš 52 užd. }

begin

  ...

end;



function plotas (k: koordinatės; sk: integer): integer;

{ randa laužtės plotą }

  var fig,             { figūros plotas }

      i, j, pradžia,

      miny, maxy,      { pirmojo ir paskutiniojo sluoksnių numeriai }

      s: integer;      { vertikalių atkarpų skaičius }

      atk: atkarpos;   { vertikalios atkarpos }

      prikl: boolean;  { ar nagrinėjamas langelis priklauso figūrai }

begin

  sudaryti_atkarpas (k, sk, atk, s);

  rikiuoti (atk, s);

  { surasime sluoksnių skaičių }

  miny := maxint; maxy := 0;

  for i := 1 to sk do

    begin

      if miny > k[i].y

         then miny := k[i].y;

      if maxy < k[i].y

         then maxy := k[i].y;

    end;

  fig := 0;

  { tirsime kiekvieną sluoksnį }

  for i := miny to maxy do

    begin

      prikl := false; { dar neradome langelio, priklausančio figūrai }

      for j := 1 to s do

        if (atk[j].pr.y < i) and (i <= atk[j].pb.y)

        then begin

               if prikl

               then fig := fig + (atk[j].pr.x - pradžia)

               else pradžia := atk[j].pr.x;

               prikl := not prikl

             end;

    end;

  plotas := fig

end; { plotas }



Testo nr.�Paaiškinimai�Rezultatas��1�Duotoji laužtė yra vienetinis kvadratas�1��2�Laužtė yra kryžiaus formos (50.1 pav.)�5��3�Laužtė yra apverstos U formos (50.2 pav.)�8��4�Duotoji laužtė – stačiakampis�15��5�Sudėtingesnė laužtė (50.3 pav.)�38��

50.3 pav. 

50.4 pav.

50.5 pav.



III etapas

51. Atsiskaitymas skolomis. Nagrinėkime sąlygoje pateiktą pavyzdį:

Pradinės skolos:                                Skolos po perskirstymo

a b 100,00                                     a b 25,00

b c 50,00                                       c b 25,00

c a 75,00

Matome, kad po skolų perskirstymo įmonė c lieka skolinga įmonei b, nors pradžioje to nebuvo.

Sudarykime fiktyvų banką paskoloms teikti. Įmonė a yra skolinga įmonei b 100 litų. Taigi, įmonė a paima iš banko fiktyvią šimto litų paskolą ir tuos pinigus perveda įmonei b. Tačiau bankas yra fiktyvus – jis pinigų neturi. Taigi, gauname, kad įmonė a lieka skolinga bankui 100 litų, o bankas skolingas 100 litų įmonei b. 

Taip visos įmonės per banką grąžina savo skolas. Tuomet bankas suveda skolų balansą. Pavyzdžiui, 

a:  -100,00 + 75,00 ( -25,00

b: +100,00 - 50,00 ( +50,00

c: +50,00 - 75,00 ( -25,00

Matome, kad suvedus balansą, įmonės a ir c bankui lieka skolingos po 25 litus, o bankas įmonei b skolingas 50 litų. Todėl bankas su įmone b atsiskaito įmonių a ir c skolomis.

Šiek tiek sunkumų sprendžiant uždavinį sukelia įmonių skaičiaus ribojimas – 10000. Tai reiškia, kad skaitydami pradinius duomenis negalime sudaryti lentelės, kurioje būtų saugomi abiejų įmonių kodai bei vienos įmonės skola antrajai. Lentelei 

array [1..10000] of record

                       a, b: integer;

                       suma: real

                     end;

prireiktų 8 ( 10000 ( 80000 ba ( 78 K. Taigi, skaitant pradinius duomenis reikia iš karto skaičiuoti įmonių ir banko tarpusavio skolas. Turėdami jau suskaičiuotą įmonių ir banko atsiskaitymų balansą, negalime įmonių surūšiuoti pagal skolos dydį: mat rūšiuojant reikės įsiminti įmonės numerį, o tam jau gali pritrūkti vietos. Todėl šiame uždavinyje duomenų įvedimas ir išvedimas neatskiriami nuo paties algoritmo.

program atsiskaitymas_skolomis;

  const MAX = 10000; { maksimalus įmonių skaičius }

        TIKSLUMAS = 0.00001;

  type bankas = array [0..MAX] of real;



  function didž (ban: bankas; skol: boolean): integer;

  { randa daugiausiai bankui įsiskolinusią (paskolinusią) įmonę }

    var im, i: integer;

  begin

    im := 0; { pradžioje tai bus fiktyvi įmonė }

    for i := 1 to MAX do

      if ((ban[i] < 0) xor not skol) and

         ((im = 0) or (abs(ban[im]) < abs (ban[i])))

         then im := i;

    didž := im

  end; { didž }



  function min (x, y: real): real;

  begin

    if x < y

       then min := x

       else min := y

  end; { min }



  var pr, rez: text;

      ban: bankas;

      bnk, prad, gal, suma: real;

      i, a, b: integer;

begin

  assign (pr, 'SKOLOS.DAT');

  reset (pr);

  assign (rez, 'SKOLOS.REZ');

  rewrite (rez);

  prad := 0; { pradinė skolų suma }

  for i := 0 to MAX do

    ban[i] := 0;

  { sudarysime fiktyvų banką }

  readln (pr, a, b, suma);

  while a <> 0 do { kol nesutinkamas įrašas iš nulių }

    begin

      prad := prad + suma;

      ban[a] := ban[a] - suma;  { įmonė A skolinasi iš banko }

      ban[b] := ban[b] + suma;  { įmonė B deda pinigus į banką }

      readln (pr, a, b, suma)

    end;

  close (pr);

  { rasime galutinę skolų sumą }

  gal := 0;

  for i := 1 to MAX do

    if ban[i] > 0

       then gal := gal + ban[i];

  writeln (rez, 'Pradinė skolų suma: ', prad: 9: 2);

  writeln (rez, 'Galutinė skolų suma: ', gal: 8: 2);

  writeln (rez, 'Galutinis skolų sąrašas:');

  { perskirstysime skolas }

  bnk := gal;

  while bnk > TIKSLUMAS do

    begin

      { ši įmonė daugiausia įsiskolinusi bankui }

      a := didž (ban, true);

      { ši įmonė daugiausiai pinigų padėjusi į banką }

      b := didž (ban, false); 

      suma := min (abs(ban[a]), ban[b]);

      ban[a] := ban[a] + suma;

      ban[b] := ban[b] - suma;

      bnk := bnk - suma;

      writeln (rez, a: 6, b: 6, suma: 8: 2);

    end;

   close (rez);

end.



Testo nr.�Pradinės ir galinės skolų sumos��1�Pradinė skolų suma: 1265.00

Galinė skolų suma: 565.00��2�Pradinė skolų suma: 1002.00

Galinė skolų suma: 1.00��3�Pradinė skolų suma: 150.00

Galinė skolų suma: 150.00��4�Pradinė skolų suma: 633.00

Galinė skolų suma: 158.25��5�Pradinė skolų suma: 285.00

Galinė skolų suma: 200.00��6�Pradinė skolų suma: 10000.00

Galinė skolų suma: 0.00��

Pradinius duomenis iliustruoja šie paveikslėliai (51.1.1-6 pav.):

51.1.1-6 pav.



52. Paskalio trikampis. Kiekvienoje Paskalio trikampio eilutėje skaičiai išdėstyti griežta tvarka: iki vidurio jie didėja, o nuo vidurio mažėja. Todėl duotus skaičius verta surikiuoti. Pasinaudokime burbulo algoritmu ir surikiuokime juos nemažėjimo tvarka:

const max = 100;

type skaičiai = array [1..max] of integer;



procedure rikiuoti (kiek: integer;

                    var sk: skaičiai);

{ surikiuoja skaičius nemažėjimo tvarka }

{ rikiuojama burbulo metodu }

  var i, a: integer;

begin

  i := 1;

  while i < kiek do

    if sk[i] <= sk[i + 1]

    then i := i + 1

    else begin

           a := sk[i];

           sk[i] := sk[i + 1];

           sk[i + 1] := a;

           if i > 1

              then i := i - 1

              else i := i + 1

         end

end; { rikiuoti }

Pirmąją Paskalio trikampio eilutę sudaro vienetas. Patikriname, ar tarp duotų skaičių yra vienetas, tada generuojame naują Paskalio trikampio eilutę, vėl tiriame, ar galime iš turimų skaičių sudaryti šią eilutę ir t. t. Darbą baigiame, kai tolesnę Paskalio trikampio eilutę sudaro daugiau narių negu yra pradinių duomenų, arba kai didžiausias naujai generuojamos trikampio eilutės skaičius didesnis už didžiausią iš duotų skaičių.

Žemiau pateikiame algoritmą, kuris pagal buvusią Paskalio trikampio eilutę gauna tolesnę eilutę:

procedure nauja(var eil: skaičiai; { Paskalio trikampio eilutė }

                var nr: integer    { jos numeris });

{ sudaro nr+1-ąją Paskalio trikampio eilutę }

  var i: integer;

      tarp: skaičiai;

begin

  tarp[1] := 1;

  for i := 2 to nr do

     tarp [i] := eil[i-1] + eil[i];

  nr := nr + 1;

  tarp[nr] := 1; { eilutė baigiama vienetu }

  eil := tarp

end; { nauja }



Atskirai paaiškinsime, kaip tikrinama, ar iš duotų skaičių galima sudaryti turimą Paskalio trikampio eilutę. Imamas kairiausias trikampio eilutės skaičius ir ieškoma dviejų tokių pat skaičių tarp pradinių duomenų. Jei radome, imamas kitas trikampio eilutės narys ir vėl ieškoma dviejų tokių skaičių, ir t. t. Jei Paskalio trikampio eilutė turi lyginį narių skaičių, tuo darbas ir baigiamas, jei nelyginį, ieškoma vidurinio
 trikampio eilutės skaičiaus:

function galima (eil, sk: skaičiai; 

                 kiek, nr: integer): boolean;

  var rasta: boolean;

      i, j: integer;

begin

  rasta := true;

  i := 1; { pradėsime nuo mažiausio iš turimų skaičių }

  j := 1; { ieškosime eilutės pirmojo skaičiaus }

  { ieškosime visų trikampio eilutės skaičių tarp duotųjų }

  while (j <= nr div 2) and rasta do

    begin

      while sk[i] < eil[j] do

        i := i + 1;

      rasta :=  (sk[i] = eil[j]) and (sk[i+1] = eil[j]);

      j := j + 1   { ieškosime tolesnio trikampio eilutės skaičiaus }

    end;

  { jei eilutėje nelyginis elementų skaičius, ieškosime viduriniojo nario }

  if rasta and (nr mod 2 <> 0)

     then begin

            while sk[i] < eil[j] do

              i := i + 1;

            rasta := sk[i] = eil[j]

          end;

  galima := rasta

end; { galima }

Galiausia
i
 pateikiame išbaigtą uždavinį sprendžiančią procedūrą:

const 
MAX
 = 100; { maksimalus pradinių duomenų kiekis }

type skaičiai = array [1..
MAX
] of integer;



procedure Paskalio_eilutė(sk: skaičiai; { duoti skaičiai }

                  kiek: integer;                   { kiek jų }

                  var eilutė: skaičiai; { rasta eilutė }

                  var kuri: integer { eilutės numeris });

  var nr, did, didž: integer;

      eil: skaičiai;

begin

  { skaičiai surikiuojami nemažėjimo tvarka }

  rikiuoti (kiek, sk);

  did := sk[kiek];  { didžiausias iš duotų skaičių }

  kiek := kiek + 1; { reikalinga
s
 tikrinimui, kad nebūtų išeita iš rėžių }

  sk[kiek] := 0;

  { sugeneruojama pirmoji Paskalio trikampio eilutė }

  nr := 1;

  eil[1] := 1;

  didž := 1; { didžiausias atitinkamos trikampio eilutės skaičius }

  while (didž <= did) and (nr <= kiek) do

    begin

      if galima(eil, sk, kiek, nr) 

         { jei galima sudaryti tokią eilutę }

      then begin

             eilutė := eil;

             kuri := nr

           end;

      { sudarome tolesnę eilutę }

      nauja(eil, nr);

      didž := eil[nr div 2 + nr mod 2]

    end

end; { Paskalio_eilutė }



Testo nr.�Rezultatai�Komentarai��1�NEGALIMA�Tarp duotųjų skaičių nėra vieneto��2�1 1���3�1 12 66 220 495 792 924 792 220 66 12 1�Ši Paskalio trikampio eilutė turi tiek narių, kiek yra pradinių duomenų��4�1 9 36 84 126 126 84 36 9 1�Pradinių duomenų yra daugiau nei trikampio eilutės ilgis. Eilutės narių skaičius – lyginis��5�1 8 28 56 70 56 28 8 1�Pradinių duomenų yra daugiau nei trikampio eilutės ilgis. Eilutės narių skaičius – nelyginis��

53. Metagrama. Uždavinys sprendžiamas grįžimo metodu. Imame pirmąjį pradinių duomenų žodį, jį įtraukiame į metagramų grandinę. Tarp duotų žodžių randame visas to žodžio metagramas ir jas paeiliui bandome įtraukti į tos pačios grandinės galą. Taip kartojame daugelį kartų.

Jei prie grandinės galo nebegalime prijungti nei vieno naujo žodžio, tuomet turimą grandinę palyginame su ilgiausia iki šiol rasta grandine ir jei pastaroji ilgesnė, ją įsidėmime. Taip išrinksime pačią ilgiausią grandinę. 

Perrenkant visus variantus gali tekti ne kartą tikrinti, ar du žodžiai sudaro metagramų grandinę, ar ne. Todėl patogiau tai patikrinti vieną kartą ir sudaryti dvimatę lentelę m, kurios elementas m[i, j] rodytų, ar i-
asis ir j-
asis žodžiai yra metagramos, ar ne. Beje, įtraukdami žodį į metagramų g
randinę, turėtume jį įsiminti, kad to paties žodžio į tą pačią grandinę neįtrauktume pakartotinai.

const MAX = 100; { tiek daugiausia gali būti žodžių }

type žodžiai = array [1..MAX] of string [20];

     lentelė = array [1..MAX, 1..MAX] of boolean;



{ žemiau surašyti globalieji kintamieji }

var { pradiniai duomenys }

    sk: integer;  { kiek duota žodžių }

    ž: žodžiai;   { duotieji žodžiai }



    ilgis,         { ilgiausios metagramų grandinės ilgis }

    kiek: integer; { sudaromos grandinės ilgis }

    met,           { ilgiausia metagramų grandinė }

    gr: žodžiai;   { sudaroma metagramų grandinė }

    būta: array [1.. MAX] of boolean; { ar žodis įtrauktas į grandinę }

    m: lentelė; { m[i, j] = true, jei i-
asis ir j-
asis žodžiai yra metagramos }



function sudaro(žod1, žod2: string): boolean;

{ ar šie žodžiai yra metagramos }

var i, skirt: integer;

begin

  if length(žod1) <> length(žod2)

     then sudaro := false

     else begin

            skirt := 0;

            for i := 1 to length (žod1) do

               if žod1[i] <> žod2[i]

                  then skirt := skirt + 1;

            sudaro := skirt = 1

          end

end; { sudaro }



procedure eiti (i: integer);

{ rekursinė procedūra, prijungianti naują žodį prie metagramų grandinės, }

{ besibaigiančios i-
uoju žodžiu }

  var eita: boolean;

      j: integer;

begin  { eiti }

  kiek := kiek + 1;

  gr[kiek] := ž[i]; { žodis įtraukiamas į grandinę }

  būta[i] := true;

  eita := false;

  for j := 1 to sk do

    if not būta[j] and (m[i, j])

       then begin

              eita := true;

              eiti(j)

            end;

  if not eita   { jei grandinė baigėsi }

     then if ilgis < kiek

             then begin

                    ilgis := kiek;

                    met := gr

                  end;

  { grįžtama žingsniu atgal }

  kiek := kiek - 1;

  būta[i] := false

end; { 
eiti
 }



procedure metagramų_grandinė;

{ iš duotų žodžių sudaroma ilgiausia metagramų grandinė }

  var i, j: integer;

begin { metagramų_grandinė }

  { rasime žodžius, kurie yra metagramos tarpusavyje }

  for i := 1 to sk do

    for j := 1 to sk do

      m[i, j] := sudaro (ž[i], ž[j]);

  { nė vienas žodis neįtrauktas į grandinę }

  for i := 1 to sk do

    būta[i] := false;

  { į metagramų grandinę bandome įtraukti kiekvieną žodį }

  ilgis := 1;   { kol kas trumpiausia grandinė sudaryta iš }

  met[ilgis] := ž[1];             { vieno žodžio }

  for i := 1 to sk do

    eiti(i)

end;  { 
metagramų_grandinė
 }



Testo nr.�Komentarai��1�Visi duotieji žodžiai vienodo ilgio ir sudaro vieną grandinę. Duotieji žodžiai pakankamai ilgi��2�Pradinių duomenų nedaug, tačiau ne visi žodžiai vienodo ilgio��3�Didžiausias testas – duota 30 skirtingo ilgo žodžių��4�Visi duotieji žodžiai yra vienodo ilgio ir sudaro vieną grandinę. Žodžiai trumpi��


54. Kryžiažodis. Šis uždavinys – taip pat perrinkimo uždavinys. Gan svarbu tvarkingai jį padaryti. Paieškai bus patogu, jei peržiūrėsime kryžiažodžio paveikslą ir įsidėmėsime visas tuščias vietas, į kurias galima įrašyti žodžius. Apie kiekvieną tuščią vietą reikėtų įsiminti tokią informaciją:



1) pradinio langelio koordinates;

2) tuščios vietos dydį (ilgį langeliais);

3) kryptį (horizontali ar vertikali).

O tuomet tvarkingai atliekame perrinkimą: į pirmą laisvą vietą bandome įstatyti pirmąjį žodį, jei pasiseka, į antrą laisvą vietą bandome įrašyti visus žodžius iš eilės, jei nepasiseka, į pirmąją vietą bandome įrašyti antrą žodį ir t. t., kol randame sprendinį.

Reikia atkreipti dėmesį į porą dalykų. Norint
 
nepamiršt
i
, kuris žodis jau įrašytas, kuris ne, būtina tai pasižymėti. Antra, įrašant žodžius, gali susidaryti situacija, kai tuščioje vietoje keletas raidžių jau yra užpildyta (54.1 pav.).

54.1 pav.



Tokiu atveju įrašant žodį, būtina tikrinti, ar įrašomo žodžio raidės sutampa su jau įrašytomis ir įsiminti, kurios raidės buvo įrašytos. Jei kartais eidami tokiu keliu sprendinio nerastume ir tektų grįžti
 atgal, neturėtume ištrinti anks
čiau įrašytų raidžių.



const MAX = 40;        { maksimalus kryžiažodžio dydis }

      Ž_MAX = 100;     { maksimalus žodžių skaičius }

      M_ILG = 10;      { maksimalus žodžio ilgis }

type kryž = array [1..MAX, 1..MAX] of char;

     žodynas = array [1..Z_MAX] of

                   record

                      žod: string [Ž_MAX];

                      ir: boolean   { ar įrašytas žodis }

                   end;

     kryptys = (hor, vert);

     duomenys = array [1..Ž_MAX] of     { apie tuščias vietas }

                    record  

                       x, y,     { pradinio langelio koordinatės }

                       ilgis: integer; { įrašomo žodžio ilgis }

                       kryptis: kryptys    { žodžio kryptis }

                    end;

     numeriai = array [1..Ž_MAX] of boolean;



procedure viena_vieta (var vt: duomenys; 

                       var sk: integer;

                       x, y, ilgis: integer; kr: kryptys);

{ įrašo informaciją apie vieną tuščią vietą }

begin

  sk := sk + 1;

  vt[sk].x := x;

  vt[sk].y := y;

  vt[sk].ilgis := ilgis;

  vt[sk].kryptis := kr

end;
 
{ 
viena_vieta
 }




procedure vietos (k: kryž; m, n: integer;

                  var vt: duomenys; { apie tuščias vietas }

                  var sk: integer   { kiek yra tuščių vietų });

{ surandamos tuščios vietos kryžiažodžio paveiksle }

  var i, j, ilgis: integer;

begin

  sk := 0;

  ilgis := 0;

  { rasime tuščias vietas eilutėse }

  for i := 1 to m do

    for j := 2 to n do

      if (k[i, j] = ' ') { langelis tuščias }

      then ilgis := ilgis + 1

      else if k[i, j - 1] = ' ' { prieš tai buvęs langelis tuščias }

           then begin

                  if ilgis > 1 { radome tuščią vietą }

                  then viena_vieta(vt, sk, i, j-ilgis,

                                   ilgis, hor);

                  ilgis := 0

                end;

  { rasime tuščias vietas stulpeliuose }

  for j := 1 to n do

    for i := 2 to m do

      if (k[i, j] = ' ') { langelis tuščias }

      then ilgis := ilgis+1

      else if k[i-1, j] = ' '  { aukščiau esantis langelis tuščias }

           then begin

                  if ilgis > 1 { radome tuščią vietą }

                  then viena_vieta (vt, sk, i-ilgis, j,

                                    ilgis, vert);

                  ilgis := 0

                end

end;
 { 
vietos
 }




procedure spręsti (var k: kryž;     { kryžiažodžio paveikslas }

                   m, n: integer;   { jo dydis }

                   ž: žodynas;      { žodynas }

                   kiek: integer;   { kiek jame žodžių }

                   var yra: boolean { ar yra sprendinys });

  var vt: duomenys;

      sk, ka, kur: integer;



  procedure trinti (ką, kur: integer; num: numeriai);

  { ištrina įrašytą žodį iš  kryžiažodžio }

  { globalieji kintamieji: vt ir z }

    var r, i, j, di, dj: integer;

  begin

    i := vt[kur].x;

    j := vt[kur].y;

    di := 0;        { kuria kryptimi trinsime }

    dj := 0;

    if vt[kur].kryptis = hor

       then dj := 1

       else di := 1;

    for r := 1 to length (ž[ka].žod) do { ištriname }

      begin

        if not num[r]

           then k[i, j] := ' ';

        i := i + di;

        j := j + dj

      end;

    ž[ka].ir := false

  end; { trinti }



  procedure įrašyti (ka, kur: integer;

                     var pasisekė: boolean;  { ar pasisekė įrašyti }

                     var num: numeriai { kurios raidės buvo įrašytos });

  { įrašo vieną žodį į kryžiažodį }

  { globalieji kintamieji: vt ir z }

    var r, i, j, di, dj: integer;

  begin

    pasisekė := true;

    for i := 1 to M_ILG do

      num[i] := false;

    i := vt[kur].x; { nuo šios vietos pradėsime įrašinėti }

    j := vt[kur].y;

    di := 0;        { kuria kryptimi įrašinėsime }

    dj := 0;

    if vt[kur].kryptis = hor

       then dj := 1

       else di := 1;

    for r := 1 to length(ž[ka].žod) do

      begin

        if k[i, j] = ' '

           then k[i, j] := ž[ka].žod[r]

           else begin

                  if k[i, j] <> ž[ka].žod[r]

                     then pasisekė := false;

                  num[r] := true

                end;

        i := i + di;

        j := j + dj

      end;

    if pasisekė

       then ž[ka].ir := true

       else trinti (ka, kur, num)

  end; { įrašyti }



  procedure pildyti (ka, kur: integer { kurį žodį kur rašyti });

  { rekursinė procedūra, užpildanti kryžiažodį }

  { globalieji kintamieji: vt ir z }

    var pasisekė: boolean;

        i: integer;

        num: numeriai; { kai įrašomas naujas žodis, čia įsimenamos vietos, }

                       { kuriose buvo įrašytos raidės iš kitų žodžių }

  begin

    if kur = 0

    then yra := true { pavyko užpildyti }

    else if length(ž[ka].žod) = vt[kur].ilgis { jei žodį galima įrašyti }

         then begin { įrašome žodį }

                įrašyti (ką, kur, pasisekė, num);

                if pasisekė    { įrašinėsime kitus žodžius }

                then begin

                       for i := 1 to kiek do

                           if not yra and not ž[i].ir

                              { jei žodis neįrašytas }

                              then pildyti (i, kur-1);

                       { ištriname įrašytąjį žodį }

                       if not yra

                          then trinti (ką, kur, num)

                     end

              end

  end; { pildyti }



begin { spręsti }

  { surandame laisvas vietas kryžiažodyje }

  vietos (k, m, n, vt, sk);

  yra := false;       { sprendinio dar nerasta }

  ką := 0;            { pradėsime pildyti nuo pirmo
jo
 žodžio }

  while (ką <= kiek) and not yra do

    begin

      ką := ką + 1;

      pildyti (ką, sk)

    end

end;
 { 
spręsti
 }




Testo nr.�Komentarai��1�Duoti trys vienodo ilgio žodžiai ir paveikslėlyje yra dvi tuščios vietos;��2�Pradiniai duomenys – daug vienodo ilgio žodžių��3�Pradiniai duomenys – skirtingo ilgio žodžiai��4�Sprendinio nėra��


55. Šaligatviai. Imkime du kvadratėlius A ir B. Tarkime, kad n
ė
 vienas iš šių kvadratėlių nesiekia lauko krašto. Šiuos kvadratėlius sujunkime trumpiausiu keliu (žr. 18 uždavinį „Paprastas kelias“). Toliau imkime kitus du taškus – C ir D: tarp jų visada galima nutiesti taką, nekertantį pirmojo (55.1 pav.).



55.1 pav.



Tarkime, kad tik vienas iš langelių A ir B liečia lauko kraštą. Įsitikiname, kad ir tuomet galima nutiesti antrą taką, nekertantį pirmojo (55.2 pav.).

55.2 pav.



Liko paskutinis atvejis – visi keturi langeliai liečia lauko kraštus. Kyla klausimas: kodėl visi keturi, o ne tik A ir B. Tarkime, kad langelis C neliečia lauko krašto. Tada langelius C ir D pervadiname į A, B, o langelius A, B – į C , D ir gauname aukščiau išnagrinėtą atvejį.

Tokiu atveju kelias, nutiestas iš A į B, dalija lauką į dvi dalis. Pakliūti iš vienos dalies į kitą, nekirtus pirmojo tako, – neįmanoma (55.3 pav.). 

55.3 pav.



Taigi šaligatvį 
galima 
nutiesti 
tik tada, jei taškai C ir D yra toje pačioje padal
y
to lauko pusėje (55.4 pav). 

55.4 pav.



Tai patikrinti nesunku. Jei eidami lauko kraštu 
pirma 
su
tiksime 
tašką A, po 
to 
C, tada B ir galiausiai D
, tai
 nesikertanči
us
 tak
us
 
bus 
ne
įmanoma
 
nutiesti
, t. y. neįmanoma tuomet, kai einant kraštu du 
ap
tikti taškai priklauso skirtingiems šaligatviams (55.5 pav.).

55.5 pav.



Eidami lauko kraštu pradėsime nuo kvadratėlio (1, 1) ir eisime prieš
 laikrodžio rodyklę (55.6 pav.):

pav.



Pateikiame išbaigtą algoritmą:

type kvadr = record { kvadratėlio koordinatės }

               x, y: integer

             end;

function kraštas(x, y: integer; k: kvadr): boolean;

{ patikrina, ar kvadratėlis k liečia lauko kraštą }

begin

  kraštas := (k.x = 1) or (k.x = x) or (k.y = 1) or (k.y = y)

end;
 { 
kraštas
 }




function toliau(k1, k2: kvadr): boolean;

{ tikrina, ar kvadratėlis k2 yra toliau už k1 }

{ einant prieš
 laikrodžio rodyklę }

begin

  if k1.y = 1

     then if k2.y = 1

             then toliau := k1.x < k2.x

             else toliau := true

     else if k2.y = 1

             then toliau := true

             else toliau := k1.x > k2.x

end;
 { 
toliau
 }




function galima(x, y: integer;

                a, b, c, d: kvadr): boolean;

begin

  if kraštas(x, y, a) and kraštas(x, y, b) and 

     kraštas(x, y, c) and kraštas(x, y, d)

     then { jei visi taškai liečia kraštą }

        if toliau(a, b)

           then if toliau(a, c) and toliau(c, b) xor

                   toliau(a, d) and toliau(d, a)

                   then galima := true

                   else galima := false

           else if toliau(b, c) and toliau(c, a) xor

                   toliau(b, d) and toliau(d, a)

                   then galima := true

                   else galima := false

     else galima := true

end;
 { 
galima
 }


Beje, sąlygoje nurodyti lauko dydžio ribojimai xmax ir ymax sprendimui yra visiškai nereikalingi.

Testo nr.�Pradiniai duomenys�Rezultatas�Komentarai��1�10 10

3 3

10 7

7 1

3 10�GALIMA�Taškai B ir D yra liečia lauko kraštus��2�500 500

1 1

500 500

1 500

500 1�NEGALIMA�Visi taškai yra kampuose��3�4 2

1 1

4 2

3 1

2 2�NEGALIMA�Taškai A ir B – kampuose, C ir D – liečia kraštą��4�300 12

2 6

299 6

2 3

200 2�GALIMA�Nė vienas taškas neliečia lauko krašto��5�6 6

3 6

6 4

6 2

3 1�GALIMA�Visi keturi taškai liečia kraštus��


56. Televizijos laidos. Pirmiausia pastebėsime, kad laidų pradžia ir pabaiga nėra sprendimui reikalingi duomenys, svarbu tik laidų trukmė. Taigi
 skaitydami pradinius duomenis, juos iš karto perskaičiuojame į laidų trukmes. Beje, minutės, skirtos reklamai, įsiminti nereikia, todėl visų laidų trukmę reikia sumažinti vienetu. Tai, kad kasetę galima keisti tik reklamos metu, reiškia, kad ta pati laida negali būti įrašyta dviejose kasetėse.



Uždavinys sprendžiamas perrinkimo būdu (arba grįžimo metodu). Akivaizdu, kad kasečių negali būti daugiau, nei iš viso yra laidų.

Sunumeruojame laidas nuo 1 iki kiek ir paimame kiek kasečių. Tuomet rekursiškai įrašinėjame laidas: kiekvieną laidą bandome įrašyti į kiekvieną kasetę. Kai įrašomos visos kasetės, sunaudotas kasečių skaičius palyginamas su iki šiol rastu mažiausiu kasečių skaičiumi ir darbas tęsiamas toliau, kol neišbandomi visi variantai. 

Deja, tų variantų gali būti labai daug. Ištisu perrinkimu parašyta programa veiks labai ilgai. Todėl tenka ieškoti kiek įmanoma daugiau būdų, kada galima „apkarpyti“ perrinkimą. Štai keletas pasiūlymų
 
kaip 
sumažin
ti
 perrinkimą:

akivaizdu, kad kasečių bus ne mažiau negu laidų, kurių trukmė didesnė 
kaip
 30 s
.
 
skaičius 
– iš karto galima išrinkti šias laidas ir įrašyti į atskiras kasetes;

laidas verta surikuoti nedidėjimo tvarka pagal jų trukmę:tuomet jei kurią nors laidą įrašėme į i-
ąją kasetę, o tolesnės laidos trukmė tokia pat, tai šios laidos neverta bandyti įrašinėti į (n-1) pirmųjų kasečių, o reikia pradėti nuo i-
osios kasetės;

jei iki šiol laidoms įrašyti panaudota kas kasečių, tai naują laidą reikia bandyti įrašyti į visas panaudotas kasetes ir vieną tuščią;

tarkime, kad laidą norime įrašyti į j-
ąją kasetę, kurioje jau įrašyta min minučių: to neverta daryti, jei tą pačią laidą jau bandėme įrašyti į kurią kitą kasetę, kurioje irgi įrašyta min minučių;

jei žinome, kad laidoms įrašyti tikrai užteks maž kasečių ir matome, kad dabartinei prireiks maž ar daugiau kasečių, šią rekursijos šaką verta nutraukti.

Žemiau pateikiame algoritmą. Jame nerasite pradinių duomenų įvedimo ir transformavimo (iš laidos pradžios ir pabaigos laiko į jos trukmę) ir laidų surikiavimo pagal trukmes. 
Laidų rikiavimui pagal ilgius
 panaudotas 52 uždavinyje pateiktas duomenų rikiavimo algoritmas.

const M = 30; { maksimalus laidų skaičius per parą }

type lentelė = array [1..M] of integer;



{ žemiau išvard
y
ti globalieji kintamieji }

var ilgiai,         { laidų ilgiai, surikiuoti nedidėjimo tvarka }

    k: lentelė;     { kasetės }

    kiekis,         
{ paros laidų skaičius
 }

    maž,            { mažiausias kasečių skaičius }

    kiek: integer;  { skaičius paros laidų, trumpesnių nei 30 min. }



function buvo (x: integer): boolean;

{ tikrina, ar nėra dar vienos kasetės, kuri užpildyta }

{ tiek pat kiek kasetė x }

  var gerai: boolean;

      j: integer;

begin

  gerai := false;

  for j := 1 to x-1 do

    if k[j] = k[x]

       then gerai := true;

  buvo := gerai

end;
 
{ 
buvo
 }




procedure įrašyti (laida,  { kuri laida }

                   i,      { į kurią kasetę įrašoma }

                   kas: integer { kiek kasečių jau panaudota });

{ rekursinė procedūra, įrašinėjanti laidas }

  var j,

      prad, pab,   { nuo kurios kasetės pradėti įrašinėti ir kuria baigti }

      sk: integer; { kasečių skaičius }

begin

  if laida = 0 { jei visos laidos jau įrašytos }

  then if kas < maž

       then maž := kas

       else

  else if (k[i] + ilgis[laida] < 60) and (kas < maž)

          { jei laidą į kasetę galima ir verta įrašyti }

       then begin

             { įrašome šią laidą }

             if k[i] = 0  { jei pradedame naują kasetę }

                then kas := kas + 1;

             k[i] := k[i] + ilgis[laida];

             { nuo kurios kasetės reikia pradėti rašyti likusias laidas }

             if laida <> kiek { jei sutampa šios ir praeitos laidos trukmės }

                then if ilgis[laida] = ilgis[laida+1]

                        then prad := i

                        else prad := 1

                else prad := 1;

              pab := kas + 1;

             { 
mėginsime
 įrašyti likusias laidas }

             for j := prad to pab do

               if not buvo(j)

                  then įrašyti (laida-1, j, kas);

             { atstatome buvusią reikšmę }

             k[i] := k[i] - l[laida]

            end

end;
 
{ 
įrašyti
 }




function kasetės: integer;

  var i, j: integer;

begin

  for i := 1 to kiek do

    k[i] := 0; { neužimta n
ė
 viena kasetė }

  maž := kiek;

  { iš karto į atskiras kasetes surašome laidas, ilgesnes
 nei 30 min. }

  j := 0; { pradedame nuo pirmos kasetės }

  i := 1; { ir nuo pirmos laidos }

  kiekis := kiek;

  while i <= kiekis do

    if ilgis[i] > 30

       then begin

              j := j + 1;

              k[j] := ilgis[i];

              ilgis[i] := ilgis[kiekis];

              kiekis := kiekis - 1

            end

       else i := i + 1;

  for i := 1 to kiekis+1 do

     įrašyti (kiekis, i, j)

  end;
 
{ 
kasetės
 }




Gan dažnai sprendžiant tokio tipo uždavinius daroma viena klaida – nepagrįstai taikomas vadinamasis godusis algoritmas. Jei šiam uždaviniui taikytume godųjį algoritmą, tuomet surikiavę laidas pagal jų trukmę, 
stengtumės
, pradėdami nuo ilgiausios laidos, kiek įmanoma daugiau užpildyti kasetę. Kai į kasetę niekas daugiau nebetelpa, imama nauja kasetė ir taip kartojama, kol surašomos visos laidos. Žemiau pateiktas pavyzdys rodo, kad toks sprendimas nėra teisingas:

Tarkime, kad laidų trukmės (be reklaminių minučių) yra

30, 20, 19, 18, 15, 14

Jei taikytume godųjį algoritmą, prireiktų trijų v
aizd
o kasečių:

1 kasetė                  2 kasetė           3 kasetė

30                           19                     14

20                           18

                     
 
         15

Nors iš tikrųjų užtektų ir dviejų kasečių:

1 kasetė               2 kasetė

30                            20 

15                            19

14                  
          18





Testo nr.�Rezultatas�Komentarai��1�6�8 laidos, iš jų 6 ilgesnės nei 30 min��2�4�6 laidos, iš jų 2 ilgesnės nei 30 min��3�10�Didžiausias testas – 28 laidos��4�10�21 laida��5�11�15 laidų��


57. Susijusios sritys. Tai gana lengvas uždavinys. Lentelėje pažymime juodus ir baltus langelius. Juodas langelis žymimas „J
“
, baltas – „B
“
, langelis, kuriame jau apsilankyta – „+“.



Imame pirmąjį langelį ir iš jo bandome eiti į visas keturias puses
,
 skaičiuodami
,
 kiek langelių apeita ir pažymėdami visus langelius, kuriuose buvome. Kai nebegalime niekur eiti, patikriname, ar ši sritis nėra didesnė už iki šiol rastą atitinkamos spalvos sritį ir jei taip, įsimename jos dydį bei kurio nors langelio koordinates. 

Toliau viską kartojame iš pradžių tol, kol nebeliks n
ė
 vieno neapeito langelio.

Iš kurio nors langelio bandant eiti į visas keturias puses, reikėtų tikrinti, ar neišeinama už lapo ribų. Norint to išvengti, patogu visus keturis lapo kraštus apgaubti jau apeitų langelių sritimi.

const max = 50; { popieriaus matmenys }

type lapo_dydis = array [0..max+1, 0..max+1] of char;

     sritis = record

                d,             { dydis }

                x, y: integer  { koordinatės }

              end;



procedure ieškoti (m, n: integer; { lapo matmenys }

                   lapas: lapo_dydis;

                   var j, b: sritis);

  { ieško didžiausios baltos ir juodos sričių }

  var dydis: integer; { srities dydis }

      spalva: char;   { srities spalva }



  procedure eiti (x, y : integer);

  { rekursinė procedūra, randanti vieną sritį }

  { globalieji kintamieji: dydis, spalva, lapas }

  begin

    dydis := dydis + 1;

    lapas[x, y] := '+'; { langelyje būta }

    if lapas[x+1, y] = spalva

       then eiti (x+1, y);

    if lapas[x, y+1] = spalva

       then eiti (x, y+1);

    if lapas[x-1, y] = spalva

       then eiti (x-1, y);

    if lapas[x, y-1] = spalva

       then eiti (x, y-1)

  end;
 { eiti
 }




  var eil, st: integer;

begin

  j.d := 0;

  b.d := 0;

  for eil := 1 to m do { apeisime visus langelius }

    for st := 1 to m do

        if lapas[eil, st] = 'B' { ieškosime baltos srities }

        then begin

               spalva := 'B';

               dydis := 0;

               eiti (eil, st);

               if dydis > b.d  { radome didesnę baltą sritį }

                  then begin

                         b.d := dydis;

                         b.x := eil;

                         b.y := st

                       end

             end

        else if lapas[eil, st] = 'J' { ieškosime juodos srities }

             then begin

                    spalva := 'J';

                    dydis := 0;

                    eiti (eil, st);

                    if dydis > j.d  { radome didesnę juodą sritį }

                       then begin

                              j.d := dydis;

                              j.x := eil;

                              j.y := st

                            end

                  end

end;
 
{ 
ieškoti
 }




Testo nr.�Rezultatai�Komentarai��1�DIDŽIAUSIA JUODOJI SRITIS: 117 1 1

NĖRA NEI VIENO BALTO LANGELIO�Sritis sudaryta tik iš juodų langelių��2�DIDŽIAUSIA JUODOJI SRITIS: 28 1 1

DIDŽIAUSIA BALTOJI SRITIS: 27 1 2�Yra vienintelė baltų langelių sritis ir dvi juodų langelių sritys��3�DIDŽIAUSIA JUODOJI SRITIS: 1 1 1

DIDŽIAUSIA BALTOJI SRITIS: 1 1 2�Juodi ir balti langeliai išdėstyti kaip šachmatų lentoje��4�DIDŽIAUSIA JUODOJI SRITIS: 117 1 10

DIDŽIAUSIA BALTOJI SRITIS: 62 5 1�Sudėtingas testas��

